
Knihovna EpsnetLib
 

1  ÚVOD

Knihovna  EpsnetLib  je  standardně  dodávaná  jako  součást  programovacího
prostředí Mosaic. Knihovna obsahuje funkce a funkční bloky umožňující komunikaci pro-
tokolem Epsnet v roli master na rozhraní Ethernet.

Pokud chceme funkce z knihovny EpsnetLib použít v aplikačním programu PLC, je
třeba nejprve přidat tuto knihovnu do projektu. Knihovna je dodávaná jako součást in-
stalace prostředí Mosaic od verze 2012.3.

 
Knihovna EpsnetLib není podporovaná na systémech TC-650, u systému TC700

nelze knihovnu použít s procesorovými moduly CP-7002, CP-7003 a CP-7005.

2  DATOVÉ TYPY
V knihovně EpsnetLib jsou definovány následující datové typy:

Identifikátor Typ Význam

TEpsnetCommand STRUCT Příkaz pro funkční blok fbEpsnetEth

TEpsnetCommandMas STRUCT Příkaz pro funkční blok fbEpsnetMas

TEpsnetCommands ENUM Komunikační služby

TEpsnetEthCmdNos ARRAY Čísla příkazů ke spuštění fbEpsnetMultiEth

TEpsnetLongCommand STRUCT Příkaz pro funkční blok fbEpsnetLongEth

TEpsnetZone ENUM Oblasti zápisníku

TEpsnetLongFrame STRUCT Zpráva s datovým polem

TEpsnetShortFrame STRUCT Zpráva bez datového pole

TEthEpsnetLongFrame STRUCT Zpráva s datovým polem (Ethernet)

TEthEpsnetShortFrame STRUCT Zpráva bez datového pole (Ethernet)

TEthEpsnetMaxFrame STRUCT Datové pole maximální velikosti Epsnet UDP

3  KONSTANTY
V knihovně EpsnetLib nejsou definovány žádné konstanty.

4  GLOBÁLNÍ PROMĚNNÉ
V knihovně EpsnetLib nejsou definovány žádné globální proměnné.

5  FUNKCE 

Knihovna EpsnetLib obsahuje následující funkce:

Funkce Popis

EpsnetCheckFrame Kontrola kontrolního součtu, včetně kontroly rámce

EpsnetCompleteFrame
Výpočet kontrolního součtu – zapíše do rámce včetně kon-
cového znaku

Příkazy pro fbEpsnetEth a fbEpsnetMultiEth:

EpsnetEthCmd Nastavení příkazu pro čtení nebo zápis bloku bytů

EpsnetEthCmdRW Nastavení příkazu pro čtení a zápis bloku bytů

EpsnetEthCmdWriteBool Nastavení příkazu pro zápis jednoho bitu

Příkazy pro fbEpsnetLongEth:

EpsnetEthLongCmd Nastavení příkazu pro čtení nebo zápis dlouhých bloků bytů

EpsnetEthLongCmdRW Nastavení příkazu pro čtení a zápis dlouhých bloků bytů

Příkazy pro fbEpsnetMas:

EpsnetMasCmd Nastavení příkazu pro čtení nebo zápis bloku bytů

EpsnetMasCmdRW Nastavení příkazu pro čtení a zápis bloku bytů

EpsnetMasCmdWriteBool Nastavení příkazu pro zápis jednoho bitu

5.1  Funkce EpsnetCheckFrame
Knihovna : EpsnetLib

Funkce  EpsnetCheckFrame provádí  kontrolu kontrolního součtu,  včetně kontroly
rámce zprávy protokolu EPSNET.

Popis proměnných :
Proměnná Typ Význam

VAR_IN_OUT

start BYTE První byte rámce (úvodní znak)

EpsnetCheckFrame

Návratová 
hodnota

BOOL TRUE pokud je rámec platný

5.2  Funkce EpsnetCompleteFrame
Knihovna : EpsnetLib

Funkce EpsnetCompleteFrame vypočte kontrolní součet rámce a zapíše jej na pří-
slušné místo. První byte rámce se předává na vstupu start. Uživatel musí zajistit, aby za
prvním byte bylo deklarováno dostatek prostoru pro zápis kontrolního součtu a koncového
znaku, což lze zajistit použitím proměnné typu TEpsnetLongFrame nebo pole o velikosti
255 bytů.

Popis proměnných :
Proměnná Typ Význam

VAR_IN_OUT

start BYTE První byte rámce (úvodní znak 1 nebo 2)

EpsnetCompleteFrame

Návratová 
hodnota

UINT Délka zprávy včetně kontrolního součtu a koncového zna-
ku. Nula pokud se nenajde správný úvodní znak nebo 
špatná délka

5.3  Funkce EpsnetEthCmd
Knihovna : EpsnetLib

Funkce EpsnetEthCmd slouží k nastavení jedné položky v řídícím poli příkazů pro
blok  fbEpsnetEth nebo  fbEpsnetMultiEth.  Funkce provede nastavení  příkazu pro čtení
nebo zápis bloku bytů. IP adresa cílového PLC se předává na vstupu  IpAdr. Pokud je
PLC za NAT je možné nastavit jiné než výchozí číslo UDP portu na vstupu  Port. Vstup
Address lze použít pokud má PLC systém podřízené stanice na sériových linkách v MPC
režimu s povoleným tunelováním. V ostatních případech lze nechat výchozí hodnotu 0.
Bit Read určuje zda budou data z PLC čtená (hodnota TRUE) nebo zapisována (hodnota
FALSE). 

Adresa bloku dat v paměti dotazovaného PLC je určena dvojicí vstupů Zone a In-
dex. Vstup  SizeOfD určuje délku komunikovaného bloku dat. Uživatel musí zajistit, aby
proměnná, která slouží pro zápis či čtení dat a jejíž první byte se předává na vstupu Data,
měla velikost minimálně rovnou hodnotě vstupu SizeOfD.

Odpovídající položka v poli příkazů se předává na vstupu Cmd.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

IpAdr TIPadr IP adresa

Port UINT Číslo portu

LocalPort UINT Číslo lokálního portu

Address USINT Adresa podřízené stanice

Read BOOL FALSE = zápis, TRUE = čtení

Zone TEpsnetZone Oblast zápisníku

Index UDINT Index prvního čteného byte v zápisníku

SizeOfD USINT Velikost proměnné pro čtení (max. 246) nebo zápis (max.
241) v bytech

VAR_IN_OUT

Cmd TEpsnet-
Command

Výsledný příkaz v poli

Data USINT První byte cílové/zdrojové proměnné

EpsnetEthCmd

Návratová 
hodnota

BOOL TRUE pokud jsou vstupní parametry platné

Příklad  programu  volání  funkce  EpsnetEthCmd. Příklad  realizuje  čtení  a  zápis
proměnné typu DATE_AND_TIME s pevnou pauzou mezi komunikacemi (odměřováno
časovačem CommTimer):

VAR_GLOBAL CONSTANT
  MAX_CMD1 : UINT := 2;
END_VAR

VAR_GLOBAL
  Commands1 : ARRAY [0..MAX_CMD1-1] OF TEpsnetCommand;
  CommandWrite AT Commands1[0];
  CommandRead  AT Commands1[1];
END_VAR

PROGRAM prgExampleEpsnetEthCmd
  VAR
    EpsnetEth : fbEpsnetEth;
    CommTimer : TON;
    IPAddr : TIPadr := [192,168,33,175];
    Data   : ARRAY [0..1] OF DATE_AND_TIME;
    CmdNo  : UINT;
  END_VAR

  //write blok
  EpsnetEthCmd(IpAdr := IPAddr, Read := false,
               Zone := epszR, Index := 60000,
               SizeOfD := sizeof(Data[0]), Cmd := CommandWrite,
               Data := void(Data[0])) ;
  //read blok
  EpsnetEthCmd(IPAddr, 61682, 17744, 0, true, epszR, 60100,
               sizeof(Data[1]), CommandRead, void(Data[1]));
               
  Data[0] := GetDateTime();
  
  //time between messages
  CommTimer(IN := NOT EpsnetEth.Busy, PT := T#1s);

  //communication
  EpsnetEth(Req := CommTimer.Q, MyAddress := 0,
            chanCode := ETH1_uni0,
            CmdNo := CmdNo, MaxCmd := MAX_CMD1,
            Timeout := T#1s,
            Commands := Commands1[0]);

  //next command
  IF EpsnetEth.Done OR EpsnetEth.Error THEN
    CmdNo := CmdNo + 1;
    IF CmdNo > 1 THEN
      CmdNo := 0;
    END_IF;
  END_IF;

END_PROGRAM

5.4  Funkce  EpsnetEthCmdRW
Knihovna : EpsnetLib

Funkce EpsnetEthCmdRW slouží k nastavení jedné položky v řídícím poli příkazů
pro blok fbEpsnetEth nebo fbEpsnetMultiEth. Funkce provede nastavení příkazu pro sou-
časné čtení  a zápis bloku bytů.  IP adresa cílového PLC se předává na vstupu  IpAdr.
Pokud je PLC za NAT je možné nastavit jiné než výchozí číslo UDP portu na vstupu Port.
Vstup Address lze použít pokud má PLC systém podřízené stanice na sériových linkách v
MPC režimu s povoleným tunelováním. V ostatních případech lze nechat výchozí hodnotu
nula.

Adresy bloků dat v paměti  dotazovaného PLC jsou určena dvojicí vstupů  Zone-
Read a  IndexRead  pro čtený blok a ZoneWrite a  IndexWrite.  Vstupy  SizeOfDRead a
SizeOfDWrite určují  délku  komunikovaných  bloků  dat.  Uživatel  musí  zajistit,  aby
proměnná, která slouží pro zápis dat a jejíž první byte se předává na vstupu DataWrite,
měla  velikost  minimálně  rovnou  hodnotě  vstupu  SizeOfDWrite.  Stejně  tak  platí,  že
proměnná, která slouží pro čtení dat a jejíž první byte se předává na vstupu  DataRead
musí mít velikost minimálně rovnou hodnotě vstupu SizeOfDRead.

Odpovídající položka v poli příkazů se předává na vstupu Cmd.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

IpAdr TIPadr IP adresa

Port UINT Číslo portu

LocalPort UINT Číslo lokálního portu

Address USINT Adresa podřízené stanice

ZoneRead TEpsnetZone Oblast zápisníku

IndexRead UDINT Index prvního čteného byte v zápisníku

SizeOfDRead USINT Velikost proměnné pro čtení (max. 246)

ZoneWrite TEpsnetZone Oblast zápisníku

IndexWrite UDINT Index prvního čteného byte v zápisníku

SizeOfDWrite USINT Velikost proměnné pro zápis (max. 237) v bytech

VAR_IN_OUT

Cmd TEpsnet-
Command

Výsledný příkaz v poli

DataRead USINT První byte cílové proměnné

DataWrite USINT První byte zdrojové proměnné

EpsnetEthCmdRW

Návratová 
hodnota

BOOL TRUE pokud jsou vstupní parametry platné

Příklad programu volání funkce  EpsnetEthCmdRW.  Program stejně jako příklad
pro  EpsnetEthCmd  realizuje čtení a zápis proměnné typu DATE_AND_TIME s pevnou
pauzou mezi komunikacemi (odměřováno časovačem  CommTimer). Na rozdíl od před-
chozímu příkladu v tomto případě proběhne čtení i zápis najednou:

VAR_GLOBAL
  CommandRW : TEpsnetCommand;
END_VAR

PROGRAM prgExampleEpsnetEthCmdRW
  VAR
    EpsnetEth : fbEpsnetEth;
    CommTimer : TON;
    IPAddr : TIPadr := [192,168,33,175];
    Data   : ARRAY [0..1] OF DATE_AND_TIME;
  END_VAR

  EpsnetEthCmdRW(IpAdr := IPAddr,
                 ZoneRead := epszR,
                 IndexRead := 50100,
                 SizeOfDRead := sizeof(Data[1]),
                 ZoneWrite := epszR,
                 IndexWrite := 50000,
                 SizeOfDWrite :=  sizeof(Data[0]),
                 Cmd := CommandRW,
                 DataRead := void(Data[1]),
                 DataWrite := void(Data[0]));

  Data[0] := GetDateTime();
  
  //time between messages
  CommTimer(IN := NOT EpsnetEth.Busy, PT := T#1s);

  //communication
  EpsnetEth(Req := CommTimer.Q, MyAddress := 0,
            chanCode := ETH1_uni1,
            CmdNo := 0, MaxCmd := 1,
            Timeout := T#1s,
            Commands := CommandRW);

END_PROGRAM

5.5  Funkce  EpsnetEthCmdWriteBool
Knihovna : EpsnetLib

Funkce  EpsnetEthCmdWriteBool slouží k nastavení  jedné položky v řídícím poli
příkazů pro blok fbEpsnetEth nebo fbEpsnetMultiEth. Funkce provede nastavení příkazu
pro zápis jednoho bitu. Tento zápis neovlivní hodnoty ostatních bitů v bytu, do kterého
patří.

IP adresa cílového PLC se předává na vstupu  IpAdr.  Pokud je PLC za NAT je
možné nastavit jiné než výchozí číslo UDP portu na vstupu Port. Vstup Address lze použít
pokud má PLC systém podřízené stanice na sériových linkách v MPC režimu s povo-
leným tunelováním. V ostatních případech lze nechat výchozí hodnotu 0.

Adresa zapisovaného  bitu  v  paměti  dotazovaného  PLC je  určena trojicí  vstupů
Zone. Index a BitIndx. Vstup BitIndx je indexem bitu v rámci byte a může nabývat hodnot
0 až 7. Uživatel musí zajistit, aby proměnná, která se předává na vstupu Data, byla typu
BOOL.

Odpovídající položka v poli příkazů se předává na vstupu Cmd.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

IpAdr TIPadr IP adresa

Port UINT Číslo portu

LocalPort UINT Číslo lokálního portu

Address USINT Adresa podřízené stanice

Zone TEpsnetZone Oblast zápisníku

Index UDINT Index prvního čteného byte v zápisníku

BitIndx USINT Index bitu v byte (0 až 7)

VAR_IN_OUT

Cmd TEpsnet-
Command

Výsledný příkaz v poli

Data USINT Zdrojová proměnná (musí být typu BOOL!)

EpsnetEthCmd

Návratová 
hodnota

BOOL TRUE pokud jsou vstupní parametry platné

Příklad programu s voláním funkce EpsnetEthCmdWriteBool. V příkladu se zapisují
jednotlivé bity a zpětně se čte hodnota bytu, do kterého zápis probíhá. Zápis je řízen troji -
cí proměnných SetBit – odstartuje zápis a čtení, BitIndx – určuje index bitu, který je mo-
difikován, BitValue – určuje jaká hodnota do bitu bude zapsána:

VAR_GLOBAL CONSTANT
  MAX_CMD2 : UINT := 2;
END_VAR

VAR_GLOBAL
  Commands2 : ARRAY [0..MAX_CMD2-1] OF TEpsnetCommand;
  CommandReadByte  AT Commands2[0];
  CommandWriteBit  AT Commands2[1];
END_VAR

PROGRAM prgExampleEpsnetEthCmdWriteBool
  VAR
    EpsnetEth : fbEpsnetEth;
    IPAddr    : TIPadr := [192,168,33,175];
    Data      : BYTE;
    SetBit    : BOOL;
    BitIndx   : USINT;
    BitValue  : BOOL;
    CmdNo     : UINT := 1;
    ErrCnt    : UDINT;
  END_VAR

  //read byte
  EpsnetEthCmd(IPAddr, 61682, 17744, 0, true, epszR, 40000,
               sizeof(Data), CommandReadByte, void(Data));
  //write bit
  EpsnetEthCmdWriteBool(IpAdr := IPAddr,
               Zone := epszR, Index := 40000,
               BitIndx := BitIndx,
               Cmd := CommandWriteBit,
               Data := BitValue);

  //communication
  EpsnetEth(Req := SetBit AND NOT EpsnetEth.Busy,
            MyAddress := 0,
            chanCode := ETH1_uni2,
            CmdNo := CmdNo, MaxCmd := MAX_CMD2,
            Timeout := T#1s,
            Commands := Commands2[0]);

  //next command
  IF EpsnetEth.Done OR EpsnetEth.Error THEN
    IF CmdNo = 1 THEN
      CmdNo := 0;
    ELSE
      CmdNo  := 1;
      SetBit := false;
    END_IF;
    IF EpsnetEth.Error THEN
      ErrCnt := ErrCnt + 1;
    END_IF;
  END_IF;
END_PROGRAM

5.6  Funkce  EpsnetEthLongCmd
Knihovna : EpsnetLib

Funkce EpsnetEthLongCmd slouží k nastavení jedné položky v řídícím poli příkazů
pro blok fbEpsnetLongEth. Funkce provede nastavení příkazu pro čtení nebo zápis bloku
bytů.  IP adresa cílového PLC se předává na vstupu  IpAdr.  Pokud je PLC za NAT je
možné nastavit jiné než výchozí číslo UDP portu na vstupu Port. Vstup Address lze použít
pokud má PLC systém podřízené stanice na sériových linkách v MPC režimu s povo-
leným tunelováním. V ostatních případech lze nechat výchozí hodnotu 0. Bit Read určuje
zda budou data z PLC čtená (hodnota TRUE) nebo zapisována (hodnota FALSE). 

Adresa bloku dat v paměti dotazovaného PLC je určena dvojicí vstupů Zone a In-
dex. Vstup  SizeOfD určuje délku komunikovaného bloku dat. Uživatel musí zajistit, aby
proměnná, která slouží pro zápis či čtení dat a jejíž první byte se předává na vstupu Data,
měla velikost minimálně rovnou hodnotě vstupu SizeOfD.

Odpovídající položka v poli příkazů se předává na vstupu Cmd.

Příklad viz  fbEpsnetLongEth

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

IpAdr TIPadr IP adresa

Port UINT číslo portu

LocalPort UINT číslo lokalního portu

Address USINT adresa podřízené stanice

Read BOOL FALSE = zápis, TRUE = čtení

Zone TEpsnetZone oblast zápisníku

Index UDINT index prvního čteného byte v zápisníku

SizeOfD UINT velikost proměnné pro čtení (max. 1230) nebo zápis 
(max. 1205)

VAR_IN_OUT

Cmd TEpsnetLong-
Command

výsledný příkaz v poli

Data USINT první byte cílové/zdrojové proměnné

EpsnetEthLongCmd

Návratová 
hodnota

BOOL nastavení příkazu pro čtení nebo zápis bloku bytů

5.7  Funkce  EpsnetEthLongCmdRW
Knihovna : EpsnetLib

Funkce  EpsnetEthLongCmdRW slouží  k  nastavení  jedné položky  v  řídícím poli
příkazů pro blok fbEpsnetLongEth. Funkce provede nastavení příkazu pro současné čtení
a zápis bloku bytů. IP adresa cílového PLC se předává na vstupu IpAdr. Pokud je PLC za
NAT je možné nastavit jiné než výchozí číslo UDP portu na vstupu Port. Vstup Address
lze použít pokud má PLC systém podřízené stanice na sériových linkách v MPC režimu s
povoleným tunelováním. V ostatních případech lze nechat výchozí hodnotu 0.

Adresy bloků dat v paměti  dotazovaného PLC jsou určena dvojicí vstupů  Zone-
Read a  IndexRead  pro čtený blok a ZoneWrite a  IndexWrite.  Vstupy  SizeOfDRead a
SizeOfDWrite určují  délku  komunikovaných  bloků  dat.  Uživatel  musí  zajistit,  aby
proměnná, která slouží pro zápis dat a jejíž první byte se předává na vstupu DataWrite,
měla  velikost  minimálně  rovnou  hodnotě  vstupu  SizeOfDWrite.  Stejně  tak  platí,  že
proměnná, která slouží pro čtení dat a jejíž první byte se předává na vstupu  DataRead
musí mít velikost minimálně rovnou hodnotě vstupu SizeOfDRead.

Odpovídající položka v poli příkazů se předává na vstupu Cmd.

Příklad viz fbEpsnetLongEth

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

IpAdr TIPadr IP adresa

Port UINT číslo portu

LocalPort UINT číslo lokalního portu

Address USINT adresa podřízené stanice

ZoneRead TEpsnetZone oblast zápisníku

IndexRead UDINT index prvního čteného byte v zápisníku

SizeOfDRead UINT velikost proměnné pro čtení (max. 1230)

ZoneWrite TEpsnetZone oblast zápisníku

IndexWrite UDINT index prvního čteného byte v zápisníku

SizeOfDWrite UINT velikost proměnné pro zápis (max. 1185)

VAR_IN_OUT

Cmd TEpsnetLong-
Command

výsledný příkaz v poli

DataRead USINT cílová/zdrojová proměnná

DataWrite USINT cílová/zdrojová proměnná

EpsnetEthLongCmdRW

Návratová 
hodnota

BOOL nastavení příkazu pro čtení a zápis bloku bytů

5.8  Funkce EpsnetMasCmd
Knihovna : EpsnetLib

Funkce EpsnetMasCmd slouží k nastavení jedné položky v řídícím poli příkazů pro
blok  fbEpsnetMas. Funkce provede nastavení příkazu pro čtení nebo zápis bloku bytů.
Vstup Address určuje adresu cílového PLC na seriové lince. Bit Read určuje zda budou
data z PLC čtená (hodnota TRUE) nebo zapisována (hodnota FALSE). 

Adresa bloku dat v paměti dotazovaného PLC je určena dvojicí vstupů Zone a In-
dex. Vstup  SizeOfD určuje délku komunikovaného bloku dat. Uživatel musí zajistit, aby
proměnná, která slouží pro zápis či čtení dat a jejíž první byte se předává na vstupu Data,
měla velikost minimálně rovnou hodnotě vstupu SizeOfD.

Odpovídající položka v poli příkazů se předává na vstupu Cmd.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Address USINT adresa podřízené stanice

Read BOOL FALSE = zápis, TRUE = čtení

Zone TEpsnetZone oblast zápisníku

Index UDINT index prvního čteného byte v zápisníku

SizeOfD USINT velikost proměnné pro čtení (max. 246) nebo zápis 
(max. 241)

VAR_IN_OUT

Cmd TEpsnetCom-
mandMas

výsledný příkaz v poli

Data USINT první byte cílové/zdrojové proměnné

EpsnetMasCmd

Návratová 
hodnota

BOOL nastavení příkazu pro čtení nebo zápis bloku bytů 
pro blok fbEpsnetMas

5.9  Funkce  EpsnetMasCmdRW
Knihovna : EpsnetLib

Funkce EpsnetMasCmdRW slouží k nastavení jedné položky v řídícím poli příkazů
pro blok  fbEpsnetMas.  Funkce provede nastavení příkazu pro současné čtení a zápis
bloku bytů. Vstup Address určuje adresu cílového PLC na seriové lince.

Adresy bloků dat v paměti  dotazovaného PLC jsou určena dvojicí vstupů  Zone-
Read a  IndexRead  pro čtený blok a ZoneWrite a  IndexWrite.  Vstupy  SizeOfDRead a
SizeOfDWrite určují  délku  komunikovaných  bloků  dat.  Uživatel  musí  zajistit,  aby
proměnná, která slouží pro zápis dat a jejíž první byte se předává na vstupu DataWrite,
měla  velikost  minimálně  rovnou  hodnotě  vstupu  SizeOfDWrite.  Stejně  tak  platí,  že
proměnná, která slouží pro čtení dat a jejíž první byte se předává na vstupu  DataRead
musí mít velikost minimálně rovnou hodnotě vstupu SizeOfDRead.

Odpovídající položka v poli příkazů se předává na vstupu Cmd.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Address USINT adresa podřízené stanice

ZoneRead TEpsnetZone oblast zápisníku

IndexRead UDINT index prvního čteného byte v zápisníku

SizeOfDRead USINT velikost proměnné pro čtení (max. 246)

ZoneWrite TEpsnetZone oblast zápisníku

IndexWrite UDINT index prvního čteného byte v zápisníku

SizeOfDWrite USINT velikost proměnné pro zápis (max. 237)

VAR_IN_OUT

Cmd TEpsnetCommand-
Mas

výsledný příkaz v poli

DataRead USINT cílová/zdrojová proměnná

DataWrite USINT cílová/zdrojová proměnná

EpsnetMasCmdRW

Návratová 
hodnota

BOOL nastavení příkazu pro čtení a zápis bloku 
bytů pro blok fbEpsnetMas

5.10  Funkce  EpsnetMasCmdWriteBool
Knihovna : EpsnetLib

Funkce  EpsnetMasCmdWriteBool slouží k nastavení jedné položky v řídícím poli
příkazů pro blok fbEpsnetMas. Funkce provede nastavení příkazu pro zápis jednoho bitu.
Tento zápis neovlivní hodnoty ostatních bitů v bytu, do kterého patří.

Vstup Address určuje adresu cílového PLC na seriové lince.
Adresa zapisovaného  bitu  v  paměti  dotazovaného  PLC je  určena trojicí  vstupů

Zone. Index a BitIndx. Vstup BitIndx je indexem bitu v rámci byte a může nabývat hodnot
0 až 7. Uživatel musí zajistit, aby proměnná, která se předává na vstupu Data, byla typu
BOOL.

Odpovídající položka v poli příkazů se předává na vstupu Cmd.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Address USINT adresa podřízené stanice

Zone TEpsnetZone oblast zápisníku

Index UDINT index prvního čteného byte v zápisníku

BitIndx USINT index bitu v byte (0 až 7)

VAR_IN_OUT

Cmd TEpsnetCom-
mandMas

výsledný příkaz v poli

Data BOOL zdrojová proměnná (musí být typu BOOL!)

EpsnetMasCmdWriteBool

Návratová 
hodnota

BOOL nastavení příkazu pro zápis jednoho bitu pro blok 
fbEpsnetMas

6  FUNKČNÍ BLOKY
V knihovně EpsnetLib je definován následující funkční blok:

Funkční blok Popis

fbEpsnetEth Zakládní blok pro komunikaci protokolem EPSNET UDP v roli nad-
řízené stanice, využívá jednu zprávu na UDP paket

fbEpsnetMultiEth Rozšířený blok fbEpsnetEth o možnost odeslat až 5 EPSNET zpráv
jedním UDP paketem

fbEpsnetLongEth Blok pro čtení a zápis delších souvislých zón. Automaticky skládá 
příkazy čtení a zápisu, aby plně využil možnosti vložit až 5 EPSNET
zpráv do jednoho UDP paketu

fbEpsnetMas Blok pro komunikaci protokolem EPSNET po sériové lince

fbEpsnetSlave Základní implementace podřízené stanice EPSNET (PC režim) na 
UNI kanálu.

6.1  Funkční blok fbEpsnetEth
Knihovna : EpsnetLib

Funkční blok  fbEpsnetEth slouží ke pro komunikaci protokolem EPSNET UDP v
pozici nadřízené stanice. Blok implementuje ze souboru služeb protokolu EPSNET násle-
dující  služby:  READN – čtení  z  datové paměti,  WRITEN – zápis do  datové paměti  ,
WANDRN – zápis do a čtení z datové paměti, WRITEB – zápis do bitů datové paměti.
Funkční blok si volí automaticky odpovídající službu podle požadovaného příkazu. Příka-
zy jsou definovány v poli, jehož první položka se předává na vstupu Cmd. Uživatel musí
zajisti, aby rozměr pole příkazů byl větší nebo roven hodnotě vstupu MaxCmd.

Jednotlivé příkazy se definují pomocí funkcí  EpsnetEthCmd, EpsnetEthCmdRW,
EpsnetEthCmdWriteBool.

Komunikace probíhá přes spojení na ethernet kanálu v režimu UNI podle konstanty
na vstupu chanCode. Spojení musí mít následujícími parametry: režim UDP, délka přijí-
mací a vysílací zóny 261 bytů. Pokud spojení není aktivní nebo nemá správné délky zón,
blok indikuje chybu na výstupech  Err hodnotou TRUE a  ErrId hodnotou 255. Adresu a
porty nastavuje blok dle parametrů jednotlivých příkazů.

Vstup  MyAddress udává zdrojovou adresu v protokolu EPSNET. Tato adresa, v
případě použití tunelování na sériové rozhraní podřízených stanic přes režim MPC, nesmí
být stejná jako adresa podřízených stanic. V případě běžné komunikace může být pone-
chána výchozí hodnota – nula.

Vykonání příkazu se spustí náběžnou hranou na vstupu Req.  Číslo příkazu, který
se má vykonat, je dáno vstupem CmdNo (prvnímu příkazu odpovídá číslo 0). Po dobu ko-
munikace  je  nastaven  výstup  Busy.  Po  ukončení  komunikace  je  nastaven  na  dobu
jednoho cyklu výstup Done v případě úspěšné komunikace nebo výstup Error v případě
chybné odpovědi nebo pokud odpověď nedorazí v čase daném vstupem Timeout. Chybu
komunikace blíže specifikuje výstup ErrId. Významy jednotlivých hodnot výstup ErrId jsou
uvedeny v tabulce popisu proměnných.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Req BOOL R_EDGE Spuštění příkazu

MyAddress USINT Vlastní adresa

chanCode UINT Kód kanálu ETH1_uni0, ..., ETH1_uni7, 
ETH2_uni0, ..., ETH2_uni7

CmdNo UINT Index příkazu v poli (0 až MaxCmd-1)

MaxCmd UINT Počet příkazů v poli

Timeout TIME Timeout příkazu

VAR_IN_OUT

Cmd TEpsnetCom-
mand

První příkaz v poli

VAR_OUTPUT

Done BOOL Nastaví se na TRUE po dobu jednoho cyklu pokud 
komunikace proběhla úspěšně

Busy BOOL TRUE znamená, že probíhá komunikace

Error BOOL Nastaví se na TRUE po dobu jednoho cyklu pokud 
nastala chyba

ErrId USINT Chybový kód:
errID = 0 bez chyby
errID = 1 odpověď nepřišla v čase Timeout
errID = 2 přijatá zpráva má chybný kontrolní součet
errID = 3 CmdNo >= MaxCmd
errID = 4 neodpovídající číslo odpovědi
errID = 5 neočekávaná odpověď
errID = 6 přijatá zpráva má neplatnou délku
errID = 12..33 chybové odpovědi z protokolu 
EPSNET
errID = 252 neplatný index bitu
errID = 253 překročena maximální délka zprávy
errID = 254 neznámá služba
errID = 255 chybné nastavení spojení na ethernet 
kanálu

Nastavení spojení na ethernetovém kanálu v režimu UNI pro funkční blok fbEpsnetEth

Příklad programu s funkčním blokem fbEpsnetEth. Program realizuje distribuci dat
do tří podřízených systémů. Zpětně je vyčítán čítač sekund ze systémových registrů. V
příkladu je použita proměnná z knihovny SysLib:

VAR_GLOBAL CONSTANT
  MAX_CMD3 : UINT := 3;
END_VAR

VAR_GLOBAL
  Commands3 : ARRAY [0..MAX_CMD3-1] OF TEpsnetCommand;
  CommandDataExchange1  AT Commands3[0];
  CommandDataExchange2  AT Commands3[1];
  CommandDataExchange3  AT Commands3[2];
END_VAR

PROGRAM prgExampleEpsnetEth
  VAR
    EpsnetEth : fbEpsnetEth;
    IPAddrs   : ARRAY [0..2] OF TIPadr :=
      [[192,168,33,154],[192,168,33,175],[192,168,33,177]];
    Ticks     : ARRAY [0..2]  OF USINT;
    Data      : ARRAY [0..99] OF USINT;
    ErrCnt    : USINT;
    CmdNo     : UINT;
  END_VAR
  VAR_TEMP
  END_VAR
  
  EpsnetEthCmdRW(IpAdr := IPAddrs[0],

                 ZoneRead := epszS,
                 IndexRead := 6,
                 SizeOfDRead := 1,
                 ZoneWrite := epszR,
                 IndexWrite := 30000,
                 SizeOfDWrite := SIZEOF(Data),
                 Cmd := CommandDataExchange1,
                 DataRead := Ticks[0],
                 DataWrite := void(Data));
  
  EpsnetEthCmdRW(IpAdr := IPAddrs[1],
                 ZoneRead := epszS,
                 IndexRead := 6,
                 SizeOfDRead := 1,
                 ZoneWrite := epszR,
                 IndexWrite := 30000,
                 SizeOfDWrite := SIZEOF(Data),
                 Cmd := CommandDataExchange2,
                 DataRead := Ticks[1],
                 DataWrite := void(Data));
                 
  EpsnetEthCmdRW(IpAdr := IPAddrs[2],
                 ZoneRead := epszS,
                 IndexRead := 6,
                 SizeOfDRead := 1,
                 ZoneWrite := epszR,
                 IndexWrite := 30000,
                 SizeOfDWrite := SIZEOF(Data),
                 Cmd := CommandDataExchange3,
                 DataRead := Ticks[2],
                 DataWrite := void(Data));

  EpsnetEth(Req := System_S.R_EDGE_1SEC,
            chanCode := ETH1_uni3, CmdNo := CmdNo,
            MaxCmd := MAX_CMD3, Timeout := T#700ms,
            Commands := Commands3[0]);

  Data[0] := Ticks[0];
  Data[1] := Ticks[1];
  Data[2] := Ticks[2];

  IF EpsnetEth.Error THEN
    ErrCnt := ErrCnt + 1;
  END_IF;

  IF EpsnetEth.Done OR ErrCnt > 1 THEN
    CmdNo := CmdNo + 1;
    IF CmdNo = MAX_CMD3 THEN
      CmdNo := 0;
    END_IF;
    ErrCnt := 0;
  END_IF;

END_PROGRAM

6.2  Funkční blok fbEpsnetMultiEth
Knihovna : EpsnetLib

Funkční blok fbEpsnetMultiEth slouží ke pro komunikaci protokolem EPSNET UDP
v  pozici  nadřízené  stanice.  Blok  implementuje  stejné  služby  jako  fbEpsnetEth,  ale
umožnuje odeslat až pět příkazů najednou. K tomu využívá skládání více EPSNET zpráv
do jednoho UDP paketu. Podmínkou tedy je, aby všechny příkazy byly pro jednu IP adre-
su.

Stejně jako pro fbEpsnetEth jsou příkazy definovány v poli, jehož první položka se
předává na vstupu Cmd. Uživatel musí zajistit, aby rozměr pole příkazů byl větší nebo ro-
ven hodnotě vstupu MaxCmd.

Pro definici příkazů se používají stejné funkce EpsnetEthCmd, EpsnetEthCmdRW,
EpsnetEthCmdWriteBool.

Komunikace probíhá přes spojení na ethernet kanálu v režimu UNI podle konstanty
na vstupu chanCode. Spojení musí mít následujícími parametry: režim UDP, délka přijí-
mací a vysílací zóny 1285 bytů. Pokud spojení není aktivní nebo nemá správné délky
zón, blok indikuje chybu na výstupech Err hodnotou TRUE a ErrId hodnotou 255. Adresu
a porty nastavuje blok dle parametrů jednotlivých příkazů.

Vykonání příkazů se spustí náběžnou hranou na vstupu Req.  Čísla příkazu, které
se má vykonat, jsou dána vstupem CmdNos (prvnímu příkazu odpovídá číslo 0). Na vstu-
pu CmdNos se očekává vždy pole pěti čísel. Počet zároveň vykonaných příkazů může být
nižší a je dán vstupem  numCmd.   Po dobu komunikace je nastaven výstup  Busy.  Po
ukončení komunikace je nastaven na dobu jednoho cyklu výstup Done v případě úspěšné
komunikace nebo výstup Error v případě chybné odpovědi nebo pokud odpověď nedorazí
v čase daném vstupem  Timeout. Chybu komunikace blíže specifikuje výstup  ErrId. Vý-
znamy jednotlivých hodnot výstup ErrId jsou uvedeny v tabulce popisu proměnných.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Req BOOL
R_EDGE

Spuštění příkazů

MyAddress USINT Vlastní adresa

chanCode UINT Kód kanálu ETH1_uni0, ..., ETH1_uni7, ETH2_uni0, ..., 
ETH2_uni7

numCmd USINT Počet příkazů k vykonání z pole CmdNos

CmdNos TEpsne-
tEthCmdNos

Indexy příkazů v poli (0 až MaxCmd-1)

MaxCmd UINT Počet příkazů v poli

Timeout TIME Timeout příkazu

VAR_OUTPUT

Done BOOL Nastaví se na TRUE po dobu jednoho cyklu pokud ko-
munikace proběhla úspěšně

Busy BOOL TRUE znamená, že probíhá komunikace

Error BOOL Nastaví se na TRUE po dobu jednoho cyklu pokud na-
stala chyba

ErrId USINT Chybový kód:
errID = 0 bez chyby
errID = 1 odpověď nepřišla v čase Timeout
errID = 2 přijatá zpráva má chybný kontrolní součet
errID = 3 CmdNo >= MaxCmd
errID = 4 neodpovídající číslo odpovědi
errID = 5 neočekávaná odpověď
errID = 6 přijatá zpráva má neplatnou délku
errID = 12..33 chybové odpovědi z protokolu EPSNET
errID = 252 neplatný index bitu
errID = 253 překročena maximální délka zprávy
errID = 254 neznámá služba
errID = 255 chybné nastavení spojení na ethernet kanálu

VAR_IN_OUT

Commands TEpsnet-
Command

První příkaz v poli

Nastavení spojení na ethernetovém kanálu v režimu UNI
pro funkční blok fbEpsnetMultiEth

Příklad programu s funkčním blokem fbEpsnetMultiEth. Program realizuje distribuci
dat do dvou podřízených systémů. Ze systému s IP adresou 192.168.134.193 je jednou
zprávou vyčten obraz vstupů ze zóny registrů X a zapsán obraz výstupů do zóny registrů
Y společně se zápisem a čtením zóny registrů R. V systému s adresou 192.168.134.194
jsou čteny a zapisovány registry v zóně R a zároveň je ve čtené zóně aleternován bit.
Díky tomu, že čtení a zápis je odesláno v jednom paketu, jsou operace synchroní a al -
ternaci bitu lze v podřízeném systému využít pro detekci nového obsahu v zapisovaných
registrech.

PROGRAM prgTestEpsnetMulti
  VAR
    EpsnetMultiEth : fbEpsnetMultiEth;
  
    Commands  : ARRAY [0..7] OF TEpsnetCommand;

    ipAdr : ARRAY [0..1] OF TIPAdr :=
      [[192,168,134,193],[192,168,134,194]];
    cmdNos : ARRAY [0..1] OF TEpsnetEthCmdNos :=
      [[0,1,2,3,4],[5,6,7,0,0]];
    cmdNums : ARRAY [0..1] OF USINT :=
      [5,3];
    idx : USINT;

    imageOfX : ARRAY [0..479] OF USINT;
    imageOfY : ARRAY [0..479] OF USINT;
    dataToRead1 : ARRAY [0..119] OF USINT;
    dataToRead2 : ARRAY [0..119] OF USINT;
    dataToWrite : ARRAY [0..59] OF USINT;
    start       : BOOL := 1;
    alt         : BOOL;
    
    trueMes  : UDINT;
    falseMes : UDINT;
  END_VAR

  EpsnetEthCmd(IpAdr := ipAdr[0],
    Read := true, Zone := TEpsnetZone#epszX,
    Index := 0, SizeOfD := 240,
    Cmd := Commands[0], Data := void(imageOfX[0]));

  EpsnetEthCmd(IpAdr := ipAdr[0],
    Read := true, Zone := TEpsnetZone#epszX,
    Index := 240, SizeOfD := 240,
    Cmd := Commands[1], Data := void(imageOfX[240]));

  EpsnetEthCmd(IpAdr := ipAdr[0],
    Read := false, Zone := TEpsnetZone#epszY,
    Index := 0, SizeOfD := 240,
    Cmd := Commands[2], Data := void(imageOfY[0]));

  EpsnetEthCmd(IpAdr := ipAdr[0],
    Read := false, Zone := TEpsnetZone#epszY,
    Index := 240, SizeOfD := 240,
    Cmd := Commands[3], Data := void(imageOfY[240]));
    
  EpsnetEthCmdRW(IpAdr := ipAdr[0],
    ZoneRead := TEpsnetZone#epszR,
    IndexRead := 60000,
    SizeOfDRead := SIZEOF(dataToRead1),
    ZoneWrite := TEpsnetZone#epszR,
    IndexWrite := 61000,
    SizeOfDWrite := SIZEOF(dataToWrite),
    Cmd := Commands[4],
    DataRead := void(dataToRead1),
    DataWrite := void(dataToWrite));
    
  EpsnetEthCmd(IpAdr := ipAdr[1],
    Read := false, Zone := TEpsnetZone#epszR,
    Index := 61000, SizeOfD := SIZEOF(dataToWrite),
    Cmd := Commands[5], Data := void(dataToWrite));

  EpsnetEthCmd(IpAdr := ipAdr[1],
    Read := true, Zone := TEpsnetZone#epszR,
    Index := 60000, SizeOfD := SIZEOF(dataToRead2),
    Cmd := Commands[6], Data := void(dataToRead2));

  EpsnetEthCmdWriteBool(IpAdr := ipAdr[1],
    Zone := TEpsnetZone#epszR,
    Index := 60000, BitIndx := 0,
    Cmd := Commands[7], Data := alt);

34 TXV 003 73.01



Knihovna EpsnetLib
 

  EpsnetMultiEth(
    chanCode := ETH1_uni1,
    numCmd := cmdNums[idx],
    CmdNos := cmdNos[idx],
    MaxCmd := 8,
    Commands := Commands[0]);

  EpsnetMultiEth.Req := false;
    
  IF start OR EpsnetMultiEth.Done OR EpsnetMultiEth.Error THEN
    start := false;
    idx := idx + 1;
    IF EpsnetMultiEth.Error THEN
      falseMes := falseMes + 1;
    ELSIF EpsnetMultiEth.Done THEN
      trueMes := trueMes + 1;
      IF idx = 2 THEN
        alt := NOT dataToRead2[0].0;
        idx := 0;
      END_IF;
    END_IF;
    EpsnetMultiEth.Req := true;
  END_IF;

END_PROGRAM

6.3  Funkční blok fbEpsnetLongEth
Knihovna : EpsnetLib

Funkční blok fbEpsnetLongEth slouží ke pro komunikaci protokolem EPSNET UDP
v pozici nadřízené stanice. Blok je určen pro přenos větších datových zón a implementuje
služby pro čtení a zápis registrů, jmenovitě READN – čtení z datové paměti, WRITEN –
zápis do datové paměti a WANDRN – zápis do a čtení z datové paměti. Narozdíl od bloků
fbEpsnetEth a  fbEpsnetMultiEth položky v poli příkazů nemusí odpovídat jedné zprávě
EPSNET, ale popisují přenos jedním UDP paketem, který může obsahovat až 5 EPNET
zpráv. Výsledkem je možnost jednou komunikací číst  zóny velikosti až 1230 bytů a zapi -
sovat zóny velikosti 1205 bytů (resp. 1185 pokud je zápis kombinován se čtením).

Příkazy jsou definovány v poli, jehož první položka se předává na vstupu  Cmd.
Uživatel  musí  zajistit,  aby  rozměr  pole  příkazů  byl  větší  nebo  roven  hodnotě  vstupu
MaxCmd.

Jednotlivé  příkazy  se  definují  pomocí  funkcí  EpsnetEthLongCmd,  Epsne-
tEthCmdLongRW.

Komunikace probíhá přes spojení na ethernet kanálu v režimu UNI podle konstanty
na vstupu chanCode. Spojení musí mít následujícími parametry: režim UDP, délka přijí-
mací a vysílací zóny 1285 bytů. Pokud spojení není aktivní nebo nemá správné délky
zón, blok indikuje chybu na výstupech Err hodnotou TRUE a ErrId hodnotou 255. Adresu
a porty nastavuje blok dle parametrů jednotlivých příkazů.

Vykonání příkazu se spustí náběžnou hranou na vstupu Req.  Číslo příkazu, který
se má vykonat, je dáno vstupem CmdNo (prvnímu příkazu odpovídá číslo 0). Po dobu ko-
munikace  je  nastaven  výstup  Busy.  Po  ukončení  komunikace  je  nastaven  na  dobu
jednoho cyklu výstup Done v případě úspěšné komunikace nebo výstup Error v případě
chybné odpovědi nebo pokud odpověď nedorazí v čase daném vstupem Timeout. Chybu
komunikace blíže specifikuje výstup ErrId. Významy jednotlivých hodnot výstup ErrId jsou
uvedeny v tabulce popisu proměnných.

36 TXV 003 73.01



Knihovna EpsnetLib
 

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Req BOOL
R_EDGE

Spuštění příkazu

MyAddress USINT Vlastní adresa

chanCode UINT Kód kanálu ETH1_uni0, ..., ETH1_uni7, ETH2_uni0, ..., 
ETH2_uni7

CmdNo UINT Index příkazu v poli (0 až MaxCmd-1)

MaxCmd UINT Počet příkazů v poli

Timeout TIME Timeout příkazu

VAR_OUTPUT

Done BOOL Nastaví se na TRUE po dobu jednoho cyklu pokud ko-
munikace proběhla úspěšně

Busy BOOL TRUE znamená, že probíhá komunikace

Error BOOL Nastaví se na TRUE po dobu jednoho cyklu pokud na-
stala chyba

ErrId USINT Chybový kód:
errID = 0 bez chyby
errID = 1 odpověď nepřišla v čase Timeout
errID = 2 přijatá zpráva má chybný kontrolní součet
errID = 3 CmdNo >= MaxCmd
errID = 4 neodpovídající číslo odpovědi
errID = 5 neočekávaná odpověď
errID = 6 přijatá zpráva má neplatnou délku
errID = 12..33 chybové odpovědi z protokolu EPSNET
errID = 252 neplatný index bitu
errID = 253 překročena maximální délka zprávy
errID = 254 neznámá služba
errID = 255 chybné nastavení spojení na ethernet kanálu

VAR_IN_OUT

Commands TEpsnet-
Long-

Command

První příkaz v poli

Nastavení spojení na ethernetovém kanálu v režimu UNI
pro funkční blok fbEpsnetLongEth

Příklad programu s funkčním blokem fbEpsnetLongEth. Program realizuje distribuci
dat do dvou podřízených systémů. Ze systému s IP adresou 192.168.134.193 je jednou
zprávou vyčten obraz vstupů ze zóny registrů X a zapsán obraz výstupů do zóny registrů
Y. V systému s adresou 192.168.134.194 jsou čteny registry dvěma zprávami (velikost
přesahuje maximální délku přenositelnou jedním paketem). Společně s vyčtením druhé
části je zpětně zapisován registr s číslem inkrementovaným při každém přenosu, umožňu-
jící kontrolu komunikace na podřízené straně.

PROGRAM prgTestEpsnetLong
  VAR
    EpsnetLongEth : fbEpsnetLongEth;
  
    Commands  : ARRAY [0..2] OF TEpsnetLongCommand;

    ipAdr : ARRAY [0..1] OF TIPAdr :=
      [[192,168,134,193],[192,168,134,194]];
    imageOfX : ARRAY [0..1199] OF USINT;
    imageOfY : ARRAY [0..479] OF USINT;
    dataToRead  : ARRAY [0..1999] OF USINT;
    dataToWrite : USINT;
    start       : BOOL := 1;

    trueMes  : UDINT;
    falseMes : UDINT;
  END_VAR

  EpsnetEthLongCmdRW(IpAdr := ipAdr[0],

    ZoneRead := TEpsnetZone#epszX,
    IndexRead := 0,
    SizeOfDRead := SIZEOF(imageOfX),
    ZoneWrite := TEpsnetZone#epszY,
    IndexWrite := 0,
    SizeOfDWrite := SIZEOF(imageOfY),
    Cmd := Commands[0],
    DataRead := void(imageOfX[0]),
    DataWrite := void(imageOfY[0]));

  EpsnetEthLongCmd(IpAdr := ipAdr[1],
    Read := true, Zone := TEpsnetZone#epszR,
    Index := 60000, SizeOfD := 1230,
    Cmd := Commands[1], Data := void(dataToRead[0]));

  EpsnetEthLongCmdRW(IpAdr := ipAdr[1],
    ZoneRead := TEpsnetZone#epszR,
    IndexRead := 61230,
    SizeOfDRead := SIZEOF(dataToWrite),
    ZoneWrite := TEpsnetZone#epszR,
    IndexWrite := 60000,
    SizeOfDWrite := 770,
    Cmd := Commands[2],
    DataRead := void(dataToRead[1230]),
    DataWrite := void(dataToWrite));

  EpsnetLongEth(
    chanCode := ETH1_uni1,
    MaxCmd := 3,
    Commands := Commands[0]);

  EpsnetLongEth.Req := false;
    
  IF start OR EpsnetLongEth.Done OR EpsnetLongEth.Error THEN
    start := false;
    IF EpsnetLongEth.Error THEN
      falseMes := falseMes + 1;
    ELSIF EpsnetLongEth.Done THEN
      trueMes := trueMes + 1;
      IF EpsnetLongEth.CmdNo = 2 THEN
        dataToWrite := dataToWrite + 1;
      END_IF;
      dataToWrite := dataToWrite + 1;
    END_IF;
    EpsnetLongEth.CmdNo := EpsnetLongEth.CmdNo + 1;
    IF EpsnetLongEth.CmdNo = 3 THEN
      EpsnetLongEth.CmdNo := 0;
    END_IF;
    EpsnetLongEth.Req := true;
  END_IF;

END_PROGRAM

6.4  Funkční blok fbEpsnetMas
Knihovna : EpsnetLib

Funkční blok fbEpsnetMas slouží ke pro komunikaci protokolem EPSNET v pozici
nadřízené stanice. Blok implementuje ze souboru služeb protokolu EPSNET následující
služby: READN – čtení z datové paměti, WRITEN – zápis do datové paměti , WANDRN –
zápis do a čtení z datové paměti, WRITEB – zápis do bitů datové paměti. Funkční blok si
volí automaticky odpovídající službu podle požadovaného příkazu. Příkazy jsou definová-
ny v poli, jehož první položka se předává na vstupu Cmd. Uživatel musí zajistit, aby roz-
měr pole příkazů byl větší nebo roven hodnotě vstupu MaxCmd.

Jednotlivé příkazy se definují pomocí funkcí EpsnetMasCmd, EpsnetMasCmdRW,
EpsnetMasCmdWriteBool.

Komunikace probíhá přes spojení na seriovém kanálu v režimu UNI podle konstan-
ty na vstupu chanCode. Spojení musí mít následujícími parametry: 8 bitů, sudá parita, 1
stop bit, délka přijímací a vysílací zóny 256 bytů. Pokud spojení není aktivní nebo nemá
správné  délky  zón,  blok  indikuje  chybu  na  výstupech  Err hodnotou  TRUE  a  ErrId
hodnotou 255.

Vstup MyAddress udává zdrojovou adresu v protokolu EPSNET. Tato adresa, ne-
smí být stejná jako adresa podřízených stanic.

Vykonání příkazu se spustí náběžnou hranou na vstupu Req.  Číslo příkazu, který
se má vykonat, je dáno vstupem CmdNo (prvnímu příkazu odpovídá číslo 0). Po dobu ko-
munikace  je  nastaven  výstup  Busy.  Po  ukončení  komunikace  je  nastaven  na  dobu
jednoho cyklu výstup Done v případě úspěšné komunikace nebo výstup Error v případě
chybné odpovědi nebo pokud odpověď nedorazí v čase daném vstupem Timeout. Chybu
komunikace blíže specifikuje výstup ErrId. Významy jednotlivých hodnot výstup ErrId jsou
uvedeny v tabulce popisu proměnných.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

Req BOOL
R_EDGE

spuštění příkazu

MyAddress USINT vlastní adresa

chanCode UINT kód kanálu CH1_uni, CH2_uni,...

CmdNo UINT index příkazu v poli (0 až MaxCmd-1)

MaxCmd UINT počet příkazů v poli

Timeout TIME timeout příkazu

VAR_OUTPUT

Done BOOL komunikace proběhla úspěšně

Busy BOOL probíhá komunikace

Error BOOL nastala chyba

ErrId USINT chybový kód:
errID = 0 bez chyby
errID = 1 odpověď nepřišla v čase Timeout
errID = 2 přijatá zpráva má chybný kontrolní součet
errID = 3 CmdNo >= MaxCmd
errID = 4 neodpovídající číslo odpovědi
errID = 5 neočekávaná odpověď
errID = 6 přijatá zpráva má neplatnou délku
errID = 12..33 chybové odpovědi z protokolu 
EPSNET
errID = 252 neplatný index bitu
errID = 253 překročena maximální délka zprávy
errID = 254 neznámá služba
errID = 255 chybné nastavení spojení na sériovém 
kanálu

VAR_IN_OUT

Commands TEpsnet-
Command-

Mas

První příkaz v poli

Nastavení spojení na sériovém kanálu v režimu UNI pro funkční blok fbEpsnetMas

6.5  Funkční blok fbEpsnetSlave
Knihovna : EpsnetLib

Funkční blok fbEpsnetSlave slouží ke pro komunikaci protokolem EPSNET v pozici
podřízené stanice. 

Blok  implementuje  ze  souboru  služeb  protokolu  EPSNET  následující  služby:
CONNECT, IDENT, SETTID, READN, WRITEN, WANDRN, READB, WRITEB.

Komunikace probíhá přes spojení na sériovém kanálu v režimu UNI podle konstan-
ty na vstupu  chanCode. Spojení musí mít následujícími parametry:  8 bitů, sudá parita,
1 stop bit, délka přijímací a vysílací zóny 256 bytů. Pokud spojení není aktivní nebo nemá
správné  délky  zón,  blok  indikuje  chybu  na  výstupech  Err hodnotou  TRUE  a  ErrId
hodnotou 255.

Vstup MyAddress udává adresu v protokolu EPSNET. Tato adresa musí být v síti
EPSNET unikátní.

Výstupy trueMes a falseMes udávají počet dobrých a špatných přijatých zpráv.
Výstup Error indikuje, že došlo k chybě. 
Významy  jednotlivých  hodnot  výstup  ErrId jsou  uvedeny  v  tabulce  popisu

proměnných.

Popis proměnných :
Proměnná Typ Význam

VAR_INPUT

myAddress USINT vlastní adresa

chanCode UINT kód kanálu CH1_uni, CH2_uni,...

VAR_OUTPUT

trueMes UDINT počet dobrých zpráv

falseMes UDINT počet špatných zpráv

Error BOOL nastala chyba
13 – bloky čtené v komunikační službě READN / 
READND se nevejdou do zprávy
16 – nulová délka bloku v komunikační službě WRI-
TEN
17 – chyba parametru WRITEB
19 – chybné parametry komunikační služby 
WANDRN / WANDRND v části zápisu
20 – chybné parametry komunikační služby 
WANDRN / WANDRND v části čtení
23 – nepodporovaná služba
144 – chybný počáteční znak
152 – chyba kontrolního součtu

ErrId USINT číslo chyby

Nastavení spojení na sériového kanálu v režimu UNI pro funkční blok fbEpsnetSlave

Příklad volání funkčního bloku realizujícího podřízenou stanici s adresou 1 na séri-
ovém kanálu CH1:

PROGRAM prgExampleEpsnetSlave
  VAR
    EpsnetSlave : fbEpsnetSlave;
  END_VAR

  EpsnetSlave(myAddress := 1, chanCode := CH1_uni);
  
END_PROGRAM

Výrobce si vyhrazuje právo na změny dokumentace. Poslední aktuální vydání je k dispo-
zici na internetu www.tecomat.cz

48 TXV 003 73.01


	1 Úvod
	2 Datové typy
	3 Konstanty
	4 Globální proměnné
	5 Funkce
	5.1 Funkce EpsnetCheckFrame
	5.2 Funkce EpsnetCompleteFrame
	5.3 Funkce EpsnetEthCmd
	5.4 Funkce EpsnetEthCmdRW
	5.5 Funkce EpsnetEthCmdWriteBool
	5.6 Funkce EpsnetEthLongCmd
	5.7 Funkce EpsnetEthLongCmdRW
	5.8 Funkce EpsnetMasCmd
	5.9 Funkce EpsnetMasCmdRW
	5.10 Funkce EpsnetMasCmdWriteBool

	6 Funkční bloky
	6.1 Funkční blok fbEpsnetEth
	6.2 Funkční blok fbEpsnetMultiEth
	6.3 Funkční blok fbEpsnetLongEth
	6.4 Funkční blok fbEpsnetMas
	6.5 Funkční blok fbEpsnetSlave


